

## Is There an Art Historian in the County?

Keith J. Brown, UNC-Office of the President, Chapel Hill, NC

### ABSTRACT

Have you ever been intrigued by a table, a chart, or a map and wanted to explore deeper into the data behind it? Have you ever produced a table, a chart, or a map, and then been asked for another just like it, but with changes – looking at a different population, subpopulation, or geographic area? If you're working in a hard-copy environment, you can either modify the program to handle each new request for additional information, or you can proactively produce a report for every possible level of the data. If there are numerous requests, the former course of action wastes your time; if there's little interest in delving deeper into the data for a particular subcategory, then taking the latter course wastes a great deal of resources. As programming becomes increasingly web-oriented, a third option opens up: adding exploratory capabilities to programs and output. This course allows you to anticipate the end user's needs for further analysis of a particular item of data and to meet that need with only the click of a mouse.

### INTRODUCTION

This paper will look at two web-based techniques for allowing users to delve into your output: pop-ups (boxes containing additional information that pop up whenever a mouse is dragged over the appropriate area), and drill-downs (output areas that are linked to other web pages or that execute a program to provide additional details). To provide a concrete foundation for our illustrations, we will be looking at an application that uses data from higher education institutions in the United States, as well as population data from the 2000 census. From this data, we will be able to generate reports, tables, and maps, all with the capability of exploring deeper into the details of our data. Only the initial web page, where the user enters the most basic information about the desired output, is static; the remaining web pages are created dynamically by our program.

### THE DATA

#### MAPPING DATA

Like most applications that deal with map graphics, our program needs two types of data – location and substantive. SAS<sup>®</sup> Institute supplies the first type through two datasets that ship as part of SAS Base: MAPS.COUNTIES and SASHELP.ZIPCODE. The first gives us the unprojected coordinates of the boundaries for all 3,000+ counties in the United States, while the second pinpoints the centroid of each ZIP code area. We will use these centroids to determine the location various higher education institutions, with each institution placed at the centroid for its ZIP code. While it might be possible to locate each institution more precisely, the benefit for our program would not be outweighed by the effort.

#### SUBSTANTIVE DATA

Our application allows you to map and report three different types of substantive data – total population, total fall headcount, or degrees conferred.

- Total Population comes from the U.S. Census Bureau's 2000 headcount at the census tract level. Tracts are intended to represent between 1,000 and 8,000 persons, and each tract is contained within a county's borders. The tract boundaries are not used in this application, but the tract centroid is; thus, the entire population of a tract is assumed to be located at its centroid.
- Total Fall Headcount is supplied by the U.S. Department of Education's Integrated Postsecondary Education Data System (IPEDS) fall enrollment data. All educational institutions whose students receive federal financial aid are required to supply data through this system. For our purposes, we have excluded all proprietary schools and all institutions which do not offer at least an associate's degree. Some institutions, generally religious colleges, choose to refuse federal funds, and so exclude themselves from the reporting requirements.
- Degrees Conferred information also comes from the IPEDS collection, and covers an entire academic year, from July 1<sup>st</sup> through June 30<sup>th</sup>. The exclusions applicable to the fall headcount are also in effect for this data. Degrees are further divided by CIP codes (Classification of Instructional Programs), a six-digit number commonly used in education circles to identify a particular program or major field. Fine Arts in general would be designated by a CIP of 50.xxxx, while Art History has a CIP code of 50.0703.

## STEPPING THROUGH THE APPLICATION

### THE STATIC WEBPAGE

UNC-OP U.S. Higher Education Information System - Microsoft Internet Explorer

File Edit View Favorites Tools Help

Back Forward Stop Home Search Favorites Links EarthLink Toolbar

 **The UNIVERSITY of NORTH CAROLINA**  
A 16-campus university

**UNC-OP Strategy Development and Analysis U.S. Higher Education Information System**

Welcome to UNC-OP SDA's system for comparing higher education offerings throughout the nation.

This application will allow you to produce maps and tables showing the degrees conferred, total college and university enrollment, and total population in a given area, or compare the numbers for two separate areas.

**Geographical Comparisons**

*Required* Center of first area is located in  - Or - Zipcode of first area is:

*Optional* Center of second area is located in  - Or - Zipcode of second area is:

*Required* Please enter your initials for printing purposes:

*Problems? Questions? Comments? Suggestions? [Let us know](#)*

Done My Computer

The static webpage contains some basic required fields and optional fields that help shape the remainder of our application. The initial required field of the center of the first area will help determine the next set of questions our program will ask to aid in the customization of the output, while the initials will be used to aid in distinguishing one set of hardcopy output from another.

Since this is a mapping tool, one of the most important pieces of information is where to begin the mapping; you have the choice of selecting by named places within a state, or if you already know the specific location, by ZIP code. A second optional area allows you to do a comparison of data between two points, a particularly useful function for economic recruitment. Not surprisingly, North Carolina is the default destination for the first area; we'll specify the ZIP code for Chapel Hill as the center for this example. Since we would like to do a comparison with Portsmouth, we'll begin by selecting "Virginia" from the pull-down menu of states for the optional second area.

#### MORE CHOICES

When we submit our choices from the initial menu, we're presented with detailed options, beginning with the distance from our selected locations. Circles can be drawn around the center points at intervals from 25 miles to 1,500 miles. Since we specified a ZIP code for our first location, the only option we have for that location is "Chapel Hill". We select our second location from a pull-down menu of locations in our previously-specified state of Virginia; the default choice is the alphabetically first town, city, or educational institution, except in North Carolina, where the default is "Research Triangle Park". The final mapping choice is whether to map the areas separately or together, allowing you to determine how detailed the maps are; if no second location is specified, this choice is not presented. Since we're looking at locations in adjoining states, both of them will fit nicely on a single map; if we were comparing Chapel Hill to San Diego, we'd probably prefer a separate map for each area.

Having specified our mapping locations and options, we turn our attention to the data to be reported. Given the title of this paper, naturally we'll select degrees conferred in the fields of Art and Art Studies (CIP code 50.0701) and Art History, Criticism and Conservation (50.0703) as well as the related ones of Arts Management (50.0704) and Fine Arts Studies (50.0799). Since the number of degrees awarded annually in these fields is likely to be relatively small, we'll boost our output by selecting all the available years. We'll also look at total fall headcount enrollment for the most recent year; since IPEDS doesn't collect enrollment information for all CIP codes, can't examine the number of students in the educational pipeline to become art historians. While there's no option for it shown, we'll also take a look at population patterns from the 2000 census within the specified circles.


#### OUTPUT #1 – DRILLABLE MAPS

The maps at the heart of our output are generated by relatively simple PROC GMAP code shown below. The 'FLAG' variables are used to control whether each of our three substantive variables produces one map or separate maps for each of our two areas.

```
PROC GMAP map=work.map(where=( flag=&flg ))
    data=dum1(where=(0<=degrees and flag&flg=1)) all;
    id state county;
    format degrees degfmt.;
    choro degrees / discrete coutline=ligr cempty=white legend=legend1
    html=href
    annotate=work.anno(where=( flag=&flg ));
```

```
run;
quit;
```

This code produced the map shown below from the options we selected earlier. While Portsmouth doesn't have any institutions of higher education, its neighbor Norfolk does. From the legend below the map, we know that Norfolk produced 123 or more graduates in our selected fields (the breakpoints for the formats being determined by the data). While we could generate the same map in printed form, the interactive nature of this application allows us to explore it further simply by positioning the cursor.



The pop-up box is a product of the 'HTML=HREF' parameter in our PROC GMAP code; this parameter also allows us to drill down into the data underlying a particular county on our map. The variable 'HREF' is simply a character variable made of up a series of HTML tags passed as text strings; the tags are associated with a particular portion of the map, so that when the cursor is positioned over that area, the HTML tags are activated. To create the pop-up box, the crucial tag is called 'ALT'. For our maps, we will first build a character variable called 'ALT', and then attach it to the other HTML tags when we build 'HREF'.

Since we have three sources of data – degrees conferred, fall enrollment, and population – and (potentially) multiple degree observations for each county, we'll use the RETAIN statement along with FIRST.variable and LAST.variable processing to reduce our data to one observation per county within our range. After we merge this dataset onto one containing all the counties that we intend to map, we'll build a truncated 'ALT' variable for the counties outside the range, and then build 'HREF' for all the counties on our map. If you position the cursor on a county within the range, you'll see a display similar to the one above; if you select a county outside the 100-mile circle, you'll see only the county and state name.

```

data dum10 dum11 dum12;
set dum10;
by flag state county;
length alt $1500;
retain alt ' ' tot89 tot99 0;

    /* Initialize variables for each county */
if first.county then do; alt=trim(countynm) || ' ' || trim(fipstate(state));
                        tot89=0; tot99=0;
end;

/* 89.xxxx is a fake cipcode for totaling all the specified CIPs */
/* 99.0000 is the cipcode for all degrees granted at an institution */
/* '0d'x forces the output to a new line; '09'x is a tab character */
if cipcode='89.xxxx' then tot89=sum(of tot89,degrees);
else if cipcode='99.0000' then tot99=sum(of tot99,degrees);

/* Degree data will sort so that our totaling CIPs come last after */
/* all the specific cipcodes we have selected */
if first.county and 0<degrees then
    alt=trim(alt) || '0d'x || '*Degrees Conferred';
if cipcode ne '99.0000' and cipcode ne '89.xxxx' then
    alt=trim(alt) || '0d'x || 'Degrees in ' || trim(cipcode) || '09'x ||
        put(degrees,comma15.);
else if cipcode='89.xxxx' then
    alt=trim(alt) || '0d'x || 'Total Specified Degrees' || '09'x ||
        put(degrees,comma15.);
else if cipcode='99.0000' then do;
    if degrees then
        alt=trim(alt) || '0d'x || 'Total Degrees Conferred' || '09'x ||
            put(degrees,comma15.);

/* Enrollment data is already summarized to one observation per county */
/* The different degree levels are represented by different variables */
if enroll then
    alt=trim(alt) || '0d'x || '**Fall Enrollment' ||
        '0d'x || 'Undergraduates' || '09'x ||
        put(ugrad,comma15.) ||
        '0d'x || 'Graduates' || '09'x ||
        put(grad,comma15.) ||
        '0d'x || 'First Prof.' || '09'x ||
        put(fprof,comma15.) ||
        '0d'x || 'Total Fall Headcount' || '09'x ||
        put(enroll,comma15.)
;

/* Population is also already summarized to one observation per county */
if pop then
    alt=trim(alt) || '0d'x || '***Total Population in Range' || '09'x ||
        put(pop,comma15.);
end;

/* Keep the final observation for each county */
/* If 99.0000 was selected, we use that for the number of degrees. */
/* If not, we go with the total of the selected cipcodes. */
if last.county;
degrees=max(of tot89,tot99);
if flag=0 then output dum10;
else if flag=1 then output dum11;
else if flag=2 then output dum12;
run;

```

```

        /* Attach the ALT variable to the mapping data                                */
data dum10;
merge dummy(in=in1)
      dum10(in=in2)
;
by state county;
if flag1=0 then flag1=.;
if flag2=0 then flag2=.;
flag0=1;
array one(*) degrees enroll pop ugrad grad fprof;
do i=1 to dim(one);
  if one(i)=. then one(i)=0;
end;
drop i;

length href $ 1500;
countynm=put(compress(put(state,z2.) || put(county,z3.)), $cnty.);

        /* Build an ALT & HREF for all the counties outside the range. */
        /* This HREF is not drillable. The ALT information (county */
        /* and state name) appears when you mouseover the county. */
if in2=0 then do;
  alt=trim(countynm) || ' ' || trim(fipstate(state));
  href=' alt="' || trim(alt) || '"';
end;

        /* Use the ALT to build HREF for all the counties inside the range. */
        /* This HREF passes parameters to the program PUB.MAP1.SAS for */
        /* drilldown. When you mouseover a county, the ALT information */
        /* appears. */
else if in2 then
href='href="' ||
"&action" ||
'?_service=default&_program=pub.map1.sas' ||
'&_debug=' || trim("&_debug") ||
'&inits=' || trim("&inits") ||
'&loop=3' ||
'&state=' || put(state,z2.) ||
'&county=' || put(county,z3.) ||
'&cips=' || trim("&cips") ||
'"' ||
' alt="' || trim(alt) || '"';
run;

```

## OUTPUT #2 – DRILLABLE REPORTS

While the 'ALT' part of the 'HREF' variable makes the pop-up box possible for all the counties on our map, the HREF tag and its associated parameters play an equally vital role in making our map drillable. If the number of possible combinations of data were relatively limited, we could generate static web pages and link those to our 'HREF'; since we'd like accommodate each county in the United States and each CIP code, building a dynamic web page is preferable.

When you click on one of the counties within our 100-mile circles, the program MAP1.SAS executes with the parameters 'INITS', 'LOOP', 'STATE', 'COUNTY', and 'CIPS'. While moving the cursor over Norfolk produces the pop-up box with its summary information, clicking on that location will allow us to delve deeper into the county's higher education statistics, as shown below.

## Educational Profile for Norfolk City County VA

**Census 2000 Population: 234,403**

INSTITUTION NAME	Fall Enrollment				Degrees Awarded in CIP Codes: '50.0701','50.0703','50.0704','50.0799'			
	Academic Year				Academic Year			
	2000- 2001	2001- 2002	2002- 2003	2003- 2004	2000- 2001	2001- 2002	2002- 2003	2003- 2004
<u>Eastern Virginia Medical School</u>	699	645	646	638	-	-	-	-
<u>Johnson &amp; Wales University-Norfolk</u>	619	635	709	831	-	-	-	-
<u>Norfolk State University</u>	6,668	6,721	6,839	6,846	12	10	23	15
<u>Old Dominion University</u>	18,969	19,627	20,105	20,802	41	42	39	45
<u>Tidewater Community College</u>	20,184	21,279	21,698	23,088	-	-	-	-
<u>Virginia Wesleyan College</u>	1,421	1,412	1,396	1,429	4	2	10	4
<b>DEGREE LEVEL</b>								
<u>Associate</u>	20,803	21,914	22,407	23,919	-	-	-	-
<u>Baccalaureate</u>	20,111	20,488	20,942	21,677	57	54	72	64
<u>Master's/ Doctoral</u>	7,235	7,500	7,625	7,606	-	-	-	-
<u>First Professional</u>	411	417	419	432	-	-	-	-
<b>INSTITUTION TYPE</b>								
<u>Public Senior</u>	25,637	26,348	26,944	27,648	53	52	62	60
<u>Private Senior</u>	2,120	2,057	2,042	2,067	4	2	10	4
<u>Public Junior</u>	20,184	21,279	21,698	23,088	-	-	-	-
<u>Private Junior</u>	619	635	709	831	-	-	-	-
<b>CARNEGIE CLASS</b>								
<u>15 - Doctoral / Research Extensive</u>	18,969	19,627	20,105	20,802	41	42	39	45
<u>21 - Masters I</u>	6,668	6,721	6,839	6,846	12	10	23	15
<u>31 - Baccalaureate -- Liberal Arts</u>	1,421	1,412	1,396	1,429	4	2	10	4
<u>40 - Associates</u>	20,803	21,914	22,407	23,919	-	-	-	-
<u>52 - Medical Schools</u>	699	645	646	638	-	-	-	-
<b>GENDER</b>								
<u>Male</u>	28,717	29,885	30,692	32,176	32	37	44	48
<u>Female</u>	19,843	20,434	20,701	21,458	25	17	28	16
<b>RACE</b>								
<u>Non-Resident Alien</u>	1,200	1,317	1,226	1,090	1	3	0	0
<u>African-American</u>	15,684	16,324	16,690	17,453	19	15	29	17
<u>Native American</u>	302	276	321	296	0	1	1	1
<u>Asian-American</u>	2,251	2,303	2,359	2,471	5	3	2	3
<u>Hispanic</u>	1,201	1,391	1,439	1,712	1	0	1	1
<u>White</u>	27,827	27,622	27,966	29,213	31	32	39	42
<u>Unknown</u>	95	1,086	1,392	1,399	0	0	0	0
<b>CIP - DISCIPLINE</b>								
<u>50.0701 - Art/Art Studies, General</u>	-	-	-	-	49	50	60	58
<u>50.0703 - Art History, Criticism and Conservation</u>	-	-	-	-	8	4	12	6
<u>99.0000 - Award Level Total</u>	48,560	50,319	51,393	53,634	-	-	-	-

UNC-OP SDA/Maps.GG038.G/25JUN05

[Printable \(Adobe PDF\) version of reports](#)

As you might guess by now, the table above uses an 'HREF' variable as well, to provide for additional drilling. Clicking on Norfolk State's link produces the table below, profiling that institution, detailing the degree levels available and breakdowns by gender, race, and discipline. If, instead we selected CIP 50.0703 – Art Criticism, History, and Conservation, the program would produce a similar table detailing the degree levels, gender, and race of students in that discipline, as well as information about the types of institutions offering that major, their Carnegie classification, and the state in which they are located.

*Profile of Norfolk State University  
A Public Senior Institution in Norfolk, Norfolk City County, Virginia  
Carnegie Class: 21 - Masters I*

	Fall Enrollment				Degrees Awarded			
	ACADEMIC YEAR				ACADEMIC YEAR			
	2000 - 2001	2001 - 2002	2002 - 2003	2003 - 2004	2000 - 2001	2001 - 2002	2002 - 2003	2003 - 2004
<b>DEGREE LEVEL</b>								
Certificate	-	-	-	-	-	-	-	8
Associate	-	-	-	-	150	132	106	130
Baccalaureate	5,890	5,963	5,968	6,039	1,638	1,396	1,640	1,460
Master's/ Doctoral	778	758	871	807	368	328	390	418
<b>GENDER</b>								
Male	4,345	4,350	4,399	4,444	1,460	1,262	1,488	1,384
Female	2,323	2,371	2,440	2,402	696	594	648	632
<b>RACE</b>								
Non-Resident Alien	39	62	54	51	12	16	18	22
African-American	5,884	5,953	6,019	5,975	1,808	1,582	1,852	1,742
Asian-American	49	49	55	58	24	32	32	32
Hispanic	85	82	95	108	34	30	20	22
Native American	23	14	21	20	8	-	8	6
White	535	543	561	481	270	196	192	174
Unknown	53	18	34	153	-	-	14	18
<b>CIP - DISCIPLINE</b>								
45.0601 - Economics, General	-	-	-	-	5	-	1	1
45.0801 - History, General	-	-	-	-	14	15	-	-
45.1001 - Political Science and Government, General	-	-	-	-	27	27	34	37
45.1101 - Sociology	-	-	-	-	45	50	46	51
45.1201 - Urban Studies/Affairs	-	-	-	-	1	3	4	15
50.0101 - Visual and Performing Arts, General	-	-	-	-	1	3	1	2
50.0701 - Art/Art Studies, General	-	-	-	-	12	10	23	15
50.0901 - Music, General	-	-	-	-	10	2	7	4
50.0903 - Music Performance, General	-	-	-	-	2	3	4	5
51.0701 - Health/Health Care Administration/Management	-	-	-	-	7	7	8	8
51.0706 - Health Information/Medical Records Administration/Administrator	-	-	-	-	5	5	2	-
51.1005 - Clinical Laboratory Science/Medical Technology/Technologist	-	-	-	-	1	3	5	3
51.1601 - Nursing/Registered Nurse (RN, ASN, BSN, MSN)	-	-	-	-	111	96	82	92
52.0101 - Business/Commerce, General	-	-	-	-	70	88	86	89
52.0204 - Office Management and Supervision	-	-	-	-	3	1	-	-
52.0301 - Accounting	-	-	-	-	40	35	25	22
52.1002 - Labor and Industrial Relations	-	-	-	-	-	1	-	-
54.0101 - History, General	-	-	-	-	-	-	7	5
99.0000 - Award Level Total	6,668	6,721	6,839	6,846	1,078	928	1,068	1,008

UNC-OP SDAMaps.GG038.G/10JUL05

### **OUTPUT #3 – NON-DRILLABLE MAPS AND REPORTS**

While the SAS Output Delivery System (ODS) allows us to build interactive applications with a minimal knowledge of the actual HTML behind them, it also provides options for producing hardcopy versions of our maps and report. The maps and the final reports can be replicated in RTF and PDF formats (among others) by the same procedures that generate our browser content. Since the drillable reports make more extensive use of HTML-specific variables, their alternative versions require invoking the REPORT or TABULATE procedure a second time, after the HTML destination has been closed.

### **CONCLUSION**

What's next? In the current incarnation, our final report is dynamically produced, but static – we can't drill any further. This does not have to be the case; since many of the categories are the same as those in the previous report, we could just as well add another round of 'HREF' variables to our data, producing a perpetually self-referencing loop. For those of us who can spend an enjoyable hour bouncing from one dictionary entry to another (forgetting in the process the word we were looking up originally), the idea of discovering that Obscure State University offers Art History, then drilling into Obscure State's data to find that it offers a basket-weaving major has great appeal.

As with many, if not most, SAS programming tasks, the bulk of the work lies in preparing the data. The tools for analysis and presentation, such as ODS, SAS/IntrNet, GMAP, REPORT, and TABULATE, are already in place and waiting to be used. I hope that you will use some of the techniques discussed here as a jumping-off point to build applications to explore your own data.

### **REFERENCES**

**SUGI 29:** Paper 090-29  
Pop-Ups, Drill-Downs, and Animation  
Mike Zdeb

**SUGI 30:** Paper 137-30  
Stretching the Bounds of SAS/GRAPH Software  
Mike Zdeb and Robert Allison

### **CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Contact the author at:

Keith J. Brown  
UNC-Office of the President  
P.O. Box 2688  
Chapel Hill, NC 27515-2688  
Work Phone: 919.962.4547  
Fax: 919.962.4316  
Email: [kjb@northcarolina.edu](mailto:kjb@northcarolina.edu)  
Web: <http://www.northcarolina.edu>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.